Kalkulus I.

Járai Antal

Ezek a programok csak szemléltetésre szolgálnak

1. Halmazok

2. Számok

```
> restart;
```

2.1. Valós számok

2.1.1. Test.

2.1.2. Példa.

```
> &+(0,0):=0; &+(0,1):=1; &+(1,0):=1; &+(1,1):=0;
  &*(0,0):=0; &*(0,1):=0; &*(1,0):=0; &*(1,1):=1;
```

$$0 \mathbin{\&+} 0 := 0$$
$$0 \mathbin{\&+} 1 := 1$$
$$1 \mathbin{\&+} 0 := 1$$
$$1 \mathbin{\&+} 1 := 0$$
$$0 \mathbin{\&*} 0 := 0$$
$$0 \mathbin{\&*} 1 := 0$$
$$1 \mathbin{\&*} 0 := 0$$
$$1 \mathbin{\&*} 1 := 1 \qquad (2.1.2.1)$$

2.1.3. Példák.

```
> `&+`:=(x,y)->irem(x+y,5); `&*`:=(x,y)->irem(x*y,5); 3&+4;
  3&*4;
```

$$\&+ := (x, y) \rightarrow irem(x + y, 5)$$
$$\&* := (x, y) \rightarrow irem(x\, y, 5)$$
$$2$$
$$2 \qquad (2.1.3.1)$$

* 2.1.4. Algebrai struktúrák.

```
> isgrupoid:=proc(G::set,f::procedure) local x,y;
  for x in G do for y in G do if not f(x,y) in G then return
  false fi;
  od; od; true; end;
```

```
isgrupoid := proc(G::set, f::procedure)                                   (2.1.4.1)
    local x, y;
    for x in G do
        for y in G do
            if not in(f(x, y), G) then
                return false
            end if
        end do
    end do;
    true
end proc
```

```
> neutral:=proc(G::set,f::procedure) local x,y,s,S;
  if not isgrupoid(G,f) then return NULL fi;
  for x in G do s:=true; for y in G do
    if f(x,y)<>y or f(y,x)<>y then s:=false; break; fi;
  od; if s then return x fi; od; NULL end;

  G:={a,b,c};neutral(G,(x,y)->y);neutral(G,(x,y)->y);

  neutral({0,1,2},(x,y)->irem(x+y,3));
```

```
neutral := proc(G::set, f::procedure)
    local x, y, s, S;
    if not isgrupoid(G, f) then
        return NULL
    end if;
    for x in G do
        s := true;
        for y in G do
            if f(x, y) <> y or f(y, x) <> y then
                s := false;
                break
            end if
```

```
            end do;
            if s then
                return x
            end if
        end do;
        NULL
    end proc
```

$$G := \{a, b, c\}$$

$$0 \quad (2.1.4.2)$$

```
> issemigroup:=proc(G::set,f::procedure) local x,y,z;
  if not isgrupoid(G,f) then return false fi;
  for x in G do for y in G do for z in G do
    if f(x,f(y,z))<>f(f(x,y),z) then return false fi;
  od; od; od; true end;

  issemigroup({a,b,c},(x,y)->x);

  issemigroup({true,false},(x,y)-> x implies y);
```

```
issemigroup := proc(G::set, f::procedure)
    local x, y, z;
    if not isgrupoid(G, f) then
        return false
    end if;
    for x in G do
        for y in G do
            for z in G do
                if f(x, f(y, z)) <> f(f(x, y), z) then
                    return false
                end if
            end do
        end do
    end do;
    true
end proc
```

true

false (2.1.4.3)

```
> isgroup:=proc(G::set,f::procedure) local x,y,n,i;
  if not isgrupoid(G,f) then return false fi;
  if not issemigroup(G,f) then return false fi;
```

```
n:=neutral(G,f); if n=NULL then return false fi;
for x in G do i:=false; for y in G do
  if f(x,y)=n and f(y,x)=n then i:=true; break fi;
od; if i=false then return false fi; od; true; end;

isgroup({0,1,2},(x,y)->irem(x+y,3));
```

isgroup := **proc**(*G*::*set*, *f*::*procedure*)
  local *x*, *y*, *n*, *i*;
  if not *isgrupoid*(*G*, *f*) **then**
    return *false*
  end if;
  if not *issemigroup*(*G*, *f*) **then**
    return *false*
  end if;
  n := *neutral*(*G*, *f*);
  if *n* = *NULL* **then**
    return *false*
  end if;
  for *x* **in** *G* **do**
    i := *false*;
    for *y* **in** *G* **do**
      if *f*(*x*, *y*) = *n* **and** *f*(*y*, *x*) = *n* **then**
        i := *true*;
        break
      end if
    end do;
    if *i* = *false* **then**
      return *false*
    end if
  end do;
  true
end proc

true (2.1.4.4)

```
> iscommutative:=proc(G::set,f::procedure) local x,y;
  if not isgrupoid(G,f) then return false fi;
  for x in G do for y in G do
    if f(x,y)<>f(y,x) then return false fi;
  od; od; true; end;
```

```
iscommutative({0,1,2},(x,y)->irem(x+y,3));
```

iscommutative := **proc**(*G*::*set*, *f*::*procedure*)
local *x*, *y*;
if not *isgrupoid*(*G*, *f*) **then**
return *false*
end if;
for *x* **in** *G* **do**
for *y* **in** *G* **do**
if *f*(*x*, *y*) <> *f*(*y*, *x*) **then**
return *false*
end if
end do
end do;
true
end proc

true (2.1.4.5)

```
> isabeliangroup:=proc(G::set,f::procedure)
  isgroup(G,f) and iscommutative(G,f) end;

  iscommutative({0,1,2},(x,y)->irem(x+y,3));
```

isabeliangroup := **proc**(*G*::*set*, *f*::*procedure*)
isgroup(*G*, *f*) **and** *iscommutative*(*G*, *f*)
end proc

true (2.1.4.6)

```
> isleftdistributive:=proc(R::set,f::procedure,g::procedure)
  local x,y,z;
  if not isgrupoid(R,f) then return false fi;
  if not isgrupoid(R,g) then return false fi;
  for x in R do for y in R do for z in R do
    if g(x,f(y,z))<>f(g(x,y),g(x,z)) then return false fi;
  od; od; od; true end;
```

isleftdistributive := **proc**(*R*::*set*, *f*::*procedure*, *g*::*procedure*) (2.1.4.7)
local *x*, *y*, *z*;
if not *isgrupoid*(*R*, *f*) **then**
return *false*
end if;
if not *isgrupoid*(*R*, *g*) **then**
return *false*
end if;

```
        for x in R do
            for y in R do
                for z in R do
                    if g(x, f(y, z)) <> f(g(x, y), g(x, z)) then
                        return false
                    end if
                end do
            end do
        end do;
        true
    end proc
```

```
> isrightdistributive:=proc(R::set,f::procedure,g::procedure)
  local x,y,z;
  if not isgrupoid(R,f) then return false fi;
  if not isgrupoid(R,g) then return false fi;
  for x in R do for y in R do for z in R do
    if g(f(y,z),x)<>f(g(y,x),g(z,x)) then return false fi;
  od; od; od; true end;
```

isrightdistributive := **proc**(R::set, f::procedure, g::procedure) (2.1.4.8)

```
        local x, y, z;
        if not isgrupoid(R, f) then
            return false
        end if;
        if not isgrupoid(R, g) then
            return false
        end if;
        for x in R do
            for y in R do
                for z in R do
                    if g(f(y, z), x) <> f(g(y, x), g(z, x)) then
                        return false
                    end if
                end do
            end do
        end do;
        true
    end proc
```

```
> isring:=proc(R::set,f::procedure,g::procedure)
  isabeliangroup(R,f) and issemigroup(R,g)
```

```
and isleftdistributive(R,f,g) and isrightdistributive(R,f,
g) end;
```

$isring := \textbf{proc}(R::set, f::procedure, g::procedure)$ (2.1.4.9)

$isabeliangroup(R, f)$ **and** $issemigroup(R,$
$g)$ **and** $isleftdistributive(R, f, g)$ **and** $isrightdistributive(R, f, g)$

end proc

```
> iscommutativering:=proc(R::set,f::procedure,g::procedure)
  isring(R,f,g) and iscommutative(R,g) end;
```

$iscommutativering := \textbf{proc}(R::set, f::procedure, g::procedure)$ (2.1.4.10)

$isring(R, f, g)$ **and** $iscommutative(R, g)$

end proc

```
> isringwithunity:=proc(R::set,f::procedure,g::procedure)
  isring(R,f,g) and neutral(R,g)<>NULL end;
```

$isringwithunity := \textbf{proc}(R::set, f::procedure, g::procedure)$ (2.1.4.11)

$isring(R, f, g)$ **and** $neutral(R, g) <> NULL$

end proc

```
> isskewfield:=proc(R::set,f::procedure,g::procedure) local
  n;
  n:=neutral(R,f); if n=NULL then return false fi;
  isring(R,f,g) and isgroup(R minus {n},g) end;
```

$isskewfield := \textbf{proc}(R::set, f::procedure, g::procedure)$ (2.1.4.12)

local n;

$n := neutral(R, f)$;

if $n = NULL$ **then**

return $false$

end if;

$isring(R, f, g)$ **and** $isgroup(minus(R, \{n\}), g)$

end proc

```
> isfield:=proc(R::set,f::procedure,g::procedure) local n;
  n:=neutral(R,f); if n=NULL then return false fi;
  isring(R,f,g) and isabeliangroup(R minus {n},g) end;
```

$isfield := \textbf{proc}(R::set, f::procedure, g::procedure)$ (2.1.4.13)

local n;

$n := neutral(R, f)$;

if $n = NULL$ **then**

return $false$

end if;

$isring(R, f, g)$ **and** $isabeliangroup(minus(R, \{n\}), g)$

end proc

* 2.1.5. Példák.

```
> iff:=(x,y)->evalb((x implies y) and (y implies x));
  X:={true,false}; isabeliangroup(X,(x,y)->iff(x,y));
```

$$iff := (x, y) \rightarrow evalb((x \Rightarrow y) \textbf{ and } (y \Rightarrow x))$$

$$X := \{false, true\}$$

$$true \quad (2.1.5.1)$$

* 2.1.6. Példák.

```
> X:={a,b,c};with(combinat,powerset);P:=powerset(X);isgroup
  (P,(x,y)->symmdiff(x,y));
```

$$X := \{a, b, c\}$$

$$[powerset]$$

$$P := \{\{\}, \{a, b, c\}, \{b, c\}, \{c\}, \{a, c\}, \{a\}, \{b\}, \{a, b\}\}$$

$$true \quad (2.1.6.1)$$

* 2.1.7. Példák.

```
> isring({0},(x,y)->0,(x,y)->0);
```

$$true \quad (2.1.7.1)$$

```
> isring(P,(x,y)->symmdiff(x,y),(x,y)->{});
```

$$true \quad (2.1.7.2)$$

* 2.1.8. Példák.

```
> iscommutativering(P,(x,y)->symmdiff(x,y),(x,y)->x intersect
  y);
  isringwithunity(P,(x,y)->symmdiff(x,y),(x,y)->x intersect
  y);
```

$$true$$

$$true \quad (2.1.8.1)$$

* 2.1.9. Példák.

2.1.10. Rendezett test.

```
> abs(7.4); abs(-3); abs(0); signum(7.4); signum(-3); signum
```

```
(0);
```

$$7.4$$
$$3$$
$$0$$
$$1$$
$$-1$$
$$0 \qquad (2.1.10.1)$$

2.1.11. Tétel.

2.1.12. Példák.

2.1.13. Tétel.

2.1.14. Valós számok.

2.1.15. Természetes számok.

```
> inc:=x->x+1;0;inc(%);inc(%);inc(%);inc(%);
  dec:=x->x-1;4;dec(%);dec(%);dec(%);
```

$$inc := x \to x+1$$
$$0$$
$$1$$
$$2$$
$$3$$
$$4$$
$$dec := x \to x-1$$
$$4$$
$$3$$
$$2$$
$$1 \qquad (2.1.15.1)$$

2.1.16. Rekurziótétel.

```
> n:=16;twopower:=1;for i to n do twopower:=twopower*2; od;
```

$$n := 16$$
$$twopower := 1$$
$$twopower := 2$$
$$twopower := 4$$
$$twopower := 8$$
$$twopower := 16$$
$$twopower := 32$$

$$twopower := 64$$
$$twopower := 128$$
$$twopower := 256$$
$$twopower := 512$$
$$twopower := 1024$$
$$twopower := 2048$$
$$twopower := 4096$$
$$twopower := 8192$$
$$twopower := 16384$$
$$twopower := 32768$$
$$twopower := 65536 \qquad (2.1.16.1)$$

2.1.17. Tétel.

2.1.18. Sorozatok.

```
> i:='i';j:='j';$3..9;i^2$i=2/3..10/3;x[i]$i=3..8;{j^i$j=i..8}$i=2..4;
```

$$i := i$$
$$j := j$$
$$3, 4, 5, 6, 7, 8, 9$$
$$\frac{4}{9}, \frac{25}{9}, \frac{64}{9}$$
$$x_3, x_4, x_5, x_6, x_7, x_8$$
$$\{4, 9, 16, 25, 36, 49, 64\}, \{27, 64, 125, 216, 343, 512\}, \{256, 625, 1296, 2401, 4096\} \qquad (2.1.18.1)$$

2.1.19. Példa.

```
> cat("ab","bcc");cat("bcc","ab");evalb(%=%%);"ab"||"bcc";
```

"abbcc"

"bccab"

false

"abbcc" (2.1.19.1)

```
> with(StringTools,Generate):Generate(4,"abc");
```

["aaaa", "aaab", "aaac", "aaba", "aabb", "aabc", "aaca", "aacb", "aacc", "abaa", "abab", "abac", "abba", "abbb", "abbc", "abca", "abcb", "abcc", "acaa", "acab", "acac", "acba", "acbb", "acbc", "acca", "accb", "accc", "baaa", "baab", "baac", "baba", "babb", "babc", "baca", "bacb", "bacc", "bbaa", "bbab", "bbac", "bbba", "bbbb", (2.1.19.2) "bbbc", "bbca", "bbcb", "bbcc", "bcaa", "bcab", "bcac", "bcba", "bcbb", "bcbc", "bcca", "bccb", "bccc", "caaa", "caab", "caac", "caba", "cabb", "cabc", "caca", "cacb", "cacc", "cbaa", "cbab", "cbac", "cbba", "cbbb", "cbbc", "cbca", "cbcb", "cbcc", "ccaa", "ccab", "ccac", "ccba", "ccbb", "ccbc", "ccca", "cccb", "cccc"]

2.1.20. *Motiváció: további rekurzív definíciók.*

2.1.21. *Általános rekurzió tétel.*

2.1.22. *Fibonacci-számok.*

```
> Fib:=proc(n::nonnegint) option remember; if n<2 then n else
  Fib(dec(n))+Fib(dec(dec(n))) fi end;interface(verboseproc=
  3):

  print(Fib);

  Fib(3);print(Fib);

  Fib(7);print(Fib);
```

```
Fib := proc(n::nonnegint)
    option remember;
    if n < 2 then
        n
    else
        Fib(dec(n)) + Fib(dec(dec(n)))
    end if
end proc
proc(n::nonnegint)
    option remember;
    if n < 2 then
        n
    else
        Fib(dec(n)) + Fib(dec(dec(n)))
    end if
end proc
```

2

```
proc(n::nonnegint)
    option remember;
    if n < 2 then
        n
```

else

$Fib(dec(n)) + Fib(dec(dec(n)))$

end if

end proc

$$13$$

proc$(n::nonnegint)$ (2.1.22.1)

option *remember*;

if $n < 2$ **then**

n

else

$Fib(dec(n)) + Fib(dec(dec(n)))$

end if

end proc

2.1.23. Szorzatok és összegek.

```
> prodfrom1:=proc(x,n) if n<1 then 1 elif n=1 then x[1] else
  prodfrom1(x,dec(n))*x[n] fi end;

  prodfrom1(y,5);

  product(y[i],i=0..7);

  sum(x[j],j=4.5..8.7);
```

prodfrom1 := **proc**(x, n)

if $n < 1$ **then**

1

elif$n = 1$ **then**

$x[1]$

else

$prodfrom1(x, dec(n)) * x[n]$

end if

end proc

$$y_1\, y_2\, y_3\, y_4\, y_5$$

$$y_0\, y_1\, y_2\, y_3\, y_4\, y_5\, y_6\, y_7$$

$$x_5 + x_6 + x_7 + x_8 \quad (2.1.23.1)$$

2.1.24. Az általános disztributivitás tétele.

```
> A:=sum(a[i],i=1..4);B:=sum(b[j],j=1..5);A*B;expand(%);
```

$$A := a_1 + a_2 + a_3 + a_4$$

$$B := b_1 + b_2 + b_3 + b_4 + b_5$$

$$(a_1 + a_2 + a_3 + a_4)(b_1 + b_2 + b_3 + b_4 + b_5)$$

$$a_1 b_1 + a_1 b_2 + a_1 b_3 + a_1 b_4 + a_1 b_5 + a_2 b_1 + a_2 b_2 + a_2 b_3 + a_2 b_4 + a_2 b_5 + a_3 b_1 + a_3 b_2 + a_3 b_3 + a_3 b_4 + a_3 b_5 + a_4 b_1 + a_4 b_2 + a_4 b_3 + a_4 b_4 + a_4 b_5 \quad (2.1.24.1)$$

2.1.25. Faktoriális, binomiális együttható.

```
> 0!; 1!; 2!; 3!; 4!; 5!; 6!;
```

1

1

2

6

24

120

720 (2.1.25.1)

```
> binomial(6,0);binomial(6,1);binomial(6,2);binomial(6,3);
  binomial(6,4);
```

1

6

15

20

15 (2.1.25.2)

2.1.26. Binomiális tétel.

```
> (x+y)^6;expand(%);
```

$$(x+y)^6$$

$$x^6 + 6x^5 y + 15x^4 y^2 + 20x^3 y^3 + 15x^2 y^4 + 6x y^5 + y^6 \quad (2.1.26.1)$$

** 2.1.27. Következmény.*

```
> sum(binomial(n,k),k=0..n);sum(binomial(n,k)*(-1)^k,k=0..n);
```

65536

0 (2.1.27.1)

2.1.28. Egész számok.

```
> type(5,integer); type(-3,integer); type(0,integer); type
  (3.14,integer);

  type(5,posint); type(-3,negint); type(0,posint); type(0,
  nonnegint); type(0,nonposint);
```

$$true$$
$$true$$
$$true$$
$$false$$
$$true$$
$$true$$
$$false$$
$$true$$
$$true \qquad (2.1.28.1)$$

2.1.29. Hatványozás egész kitevővel.

```
> x^(-5); x^(n+m); expand(%); (x^4)^(-5); (x*y)^5;
```

$$\frac{1}{x^5}$$
$$x^{16+m}$$
$$x^{16}\,x^m$$
$$\frac{1}{x^{20}}$$
$$x^5\,y^5 \qquad (2.1.29.1)$$

2.1.30. Racionális számok.

```
> type(5/7,rational); type(0,rational);
```

$$true$$
$$true \qquad (2.1.30.1)$$

2.1.31. Archimédészi tulajdonság.

2.1.32. Állítás.

2.1.33. Egész rész, maradék.

```
> floor(3.14); ceil(3.14); ceil(-3.14);
```

$$3$$
$$4$$
$$-3 \quad (2.1.33.1)$$

```
> Rmod:=proc(x::realcons,y::realcons) if y=0 then x else x-floor(x/y)*y fi; end;

  Rmod(5,0); Rmod(3.1415,2.78);
```

Rmod := **proc**(*x*::*realcons*, *y*::*realcons*)
  if *y* = 0 **then**
    x
  else
    x − floor(*x* / *y*) * *y*
  end if
end proc

$$5$$
$$0.3615 \quad (2.1.33.2)$$

2.1.34. Tétel.

2.1.35. Tétel: gyökvonás.

```
> root[2](2); evalf(%); sqrt(2); root[12](2); evalf(%);
```

$$\sqrt{2}$$
$$1.414213562$$
$$\sqrt{2}$$
$$2^{1/12}$$
$$1.059463094 \quad (2.1.35.1)$$

2.1.36. Következmény.

* 2.1.37. Állítás.

* 2.1.38. Állítás.

2.1.39. Bővített valós számok.

```
> infinity; -infinity; evalb(5<infinity); evalb(5<-infinity);
```

$$\infty$$
$$-\infty$$
true
false (2.1.39.1)

2.2. Megszámlálható halmazok.

```
> restart;
```

2.2.1. Halmazok ekvivalenciája.
2.2.2. Állítás.
2.2.3. Megjegyzés.
2.2.4. Tétel.
2.2.5. Tétel.
2.2.6. Véges és végtelen halmazok.
2.2.7. Karakterisztikus függvények.

```
> X:={a,b,c,d,e}; Y:={a,c,e}; chi:=x-> if x in Y then 1 elif
  x in X then 0 else FAIL fi;chi(a);chi(b);chi(1);
```

$$X := \{a, b, c, e, d\}$$

$$Y := \{a, c, e\}$$

$$\chi := x \rightarrow \textbf{if } x \in Y \textbf{ then } 1 \textbf{ elif } x \in X \textbf{ then } 0 \textbf{ else } FAIL \textbf{ end if}$$

$$1$$

$$0$$

$$FAIL \qquad (2.2.7.1)$$

2.2.8. Tétel.
2.2.9 Skatulya elv.
2.2.10. Tétel.
2.2.11. Megszámlálható halmazok.
2.2.12. Tétel.
2.2.13. Tétel.
2.2.14. Tétel.
2.2.15. Tétel.

```
> for k from 0 do for m from 0 to k do n:=k-m: T:=time();
  while time()<T+1 do od; print([m,n],(k*(k+1)/2)+m); od; od;
```

$$[0, 0], 0$$

$$[0, 1], 1$$

$$[1, 0], 2$$

$$[0, 2], 3$$

$[1, 1], 4$
$[2, 0], 5$
$[0, 3], 6$
$[1, 2], 7$
$[2, 1], 8$
$[3, 0], 9$
$[0, 4], 10$
$[1, 3], 11$
$[2, 2], 12$
$[3, 1], 13$

```
Warning,  computation interrupted
```

▶ ***2.2.16. Tétel.***

▶ ***2.2.17. Következmény.***

▶ ***2.2.18. Tétel.***

▶ ***2.2.19. Következmény.***

▶ ***2.2.20. Cantor tétele.***

▼ 2.3. Komplex számok

```
> restart;
```

▼ *2.3.1. Komplex számok.*

```
> `&+`:=proc(z,w) [z[1]+w[1],z[2]+w[2]] end;
  `&*`:=proc(z,w) [z[1]*w[1]-z[2]*w[2],z[1]*w[2]+z[2]*w[1]]
  end;

  [x,y]&+[0,0]; [x,y]&+[-x,-y]; [x,y]&*[1,0];

  [x,y]&*[x/(x^2+y^2),-y/(x^2+y^2)]; simplify(%);

  [0,1]&*[0,1];
```

$$\&+ := \textbf{proc}(z, w)\ [z[1]+w[1], z[2]+w[2]]\ \textbf{end proc}$$

$$\&* := \textbf{proc}(z, w)$$
$$[z[1]*w[1]-z[2]*w[2], z[1]*w[2]+z[2]*w[1]]$$
$$\textbf{end proc}$$

$[x, y]$
$[0, 0]$
$[x, y]$

$$\left[\frac{x^2}{x^2+y^2}+\frac{y^2}{x^2+y^2},\ 0\right]$$
$$[1,\ 0]$$
$$[-1,\ 0] \qquad (2.3.1.1)$$

```
> Complex(3,5); z:=3+5*I; w:=-2-6*I; z*w; Re(z); Im(z);
  conjugate(z);
```

$$3+5\,\mathrm{I}$$
$$z := 3+5\,\mathrm{I}$$
$$w := -2-6\,\mathrm{I}$$
$$24-28\,\mathrm{I}$$
$$3$$
$$5$$
$$3-5\,\mathrm{I} \qquad (2.3.1.2)$$

```
> z:='z';w:='w'; conjugate(z);

  conjugate(conjugate(z));conjugate(z+w);conjugate(1/z);
```

$$z := z$$
$$w := w$$
$$\bar{z}$$
$$\frac{z}{\overline{z+w}}$$
$$\frac{1}{\bar{z}} \qquad (2.3.1.3)$$

2.3.2. *Példa.*

```
> 64/(3^(1/2)+I); evalc(%);
```

$$\frac{64}{\sqrt{3}+\mathrm{I}}$$
$$16\sqrt{3}-16\,\mathrm{I} \qquad (2.3.2.1)$$

2.3.3. *Komplex számok abszolút értéke.*

```
> z:=x+I*y;abs(z);evalc(%);evalc(1/(x+I*y));evalc(conjugate
  (z)/abs(z)^2);
```

$$z := x+\mathrm{I}\,y$$
$$|x+\mathrm{I}\,y|$$

$$\sqrt{x^2+y^2}$$

$$\frac{x}{x^2+y^2}-\frac{\mathrm{I}\,y}{x^2+y^2}$$

$$\frac{x}{x^2+y^2}-\frac{\mathrm{I}\,y}{x^2+y^2} \qquad (2.3.3.1)$$

```
> signum(3+4*I); signum(-5); signum(0);
```

$$\frac{3}{5}+\frac{4}{5}\,\mathrm{I}$$

$$-1$$

$$0 \qquad (2.3.3.2)$$

2.3.4. Komplex számok argumentuma és trigonometrikus alakja.

```
> polar(x+I*y); op(1,%); op(2,%%); polar(3+4*I); evalc(%);
  argument(3+I*4);
```

$$\mathrm{polar}\left(|x+\mathrm{I}\,y|,\ \mathrm{argument}(x+\mathrm{I}\,y)\right)$$

$$|x+\mathrm{I}\,y|$$

$$\mathrm{argument}(x+\mathrm{I}\,y)$$

$$\mathrm{polar}\left(5,\ \arctan\left(\frac{4}{3}\right)\right)$$

$$3+4\,\mathrm{I}$$

$$\arctan\left(\frac{4}{3}\right) \qquad (2.3.4.1)$$

2.3.5. Példa.

```
> z:=16*sqrt(3)-I*16; polar(z);
```

$$z := 16\sqrt{3}-16\,\mathrm{I}$$

$$\mathrm{polar}\left(32,\ -\frac{1}{6}\,\pi\right) \qquad (2.3.5.1)$$

2.3.6. Gyökvonás komplex számból.

2.3.7. Példa.

```
> z:='z'; i:='i'; w:=16*sqrt(3)-I*16; solve(z^5=w,z); z1:=w^
  (1/5);

  r:=abs(w); phi:=argument(w);
```

```
r^(1/5)*(cos(phi/5+i*2*Pi/5)+I*sin(phi/5+i*2*Pi/5))$i=0..4;
evalf(%);

solve(z^5=1); map(z->evalf(z*z1),[%]);
```

$$z := z$$
$$i := i$$
$$w := 16\sqrt{3} - 16\,I$$

Warning, solutions may have been lost

$$z1 := \left(16\sqrt{3} - 16\,I\right)^{1/5}$$
$$r := 32$$
$$\phi := -\frac{1}{6}\pi$$

$$32^{1/5}\left(\sin\left(\frac{7}{15}\pi\right) - I\cos\left(\frac{7}{15}\pi\right)\right), 32^{1/5}\left(\sin\left(\frac{2}{15}\pi\right) + I\cos\left(\frac{2}{15}\pi\right)\right), 32^{1/5}\left(-\sin\left(\frac{4}{15}\pi\right) + I\cos\left(\frac{4}{15}\pi\right)\right), 32^{1/5}\left(-\frac{1}{2}\sqrt{3} - \frac{1}{2}I\right), 32^{1/5}\left(\sin\left(\frac{1}{15}\pi\right) - I\cos\left(\frac{1}{15}\pi\right)\right)$$

1.989043791 − 0.2090569258 I, 0.8134732860 + 1.827090915 I, −1.486289651 + 1.338261212 I, −1.732050808 − 1.000000000 I, 0.4158233818 − 1.956295201 I

Warning, solutions may have been lost

$$1, -\frac{1}{4} + \frac{1}{4}\sqrt{5} + \frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}, -\frac{1}{4} - \frac{1}{4}\sqrt{5} + \frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}, -\frac{1}{4} - \frac{1}{4}\sqrt{5} - \frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}, -\frac{1}{4} + \frac{1}{4}\sqrt{5} - \frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}$$

[1.989043791 − 0.2090569265 I, 0.8134732858 + 1.827090915 I, −1.486289651 + 1.338261212 I, −1.732050807 − 0.9999999996 I, 0.4158233815 − 1.956295201 I] (2.3.7.1)

2.3.8. *Bővített komplex számok.*

```
> z:=infinity+I*infinity; w:=infinity-I*infinity; evalb(z=w);
```

$$z := \infty + \infty\,I$$
$$w := \infty - \infty\,I$$
$$true \qquad (2.3.8.1)$$

2.3.9. Kvaterniók.

```
> `&+`:=(p,q)->[p[1]+q[1],p[2]+q[2]];
  `&*`:=(p,q)->[p[1]*q[1]-conjugate(q[2])*p[2],q[2]*p[1]+p[2]
  *conjugate(q[1])];
```

$$\&+ := (p, q) \to [p_1 + q_1, p_2 + q_2]$$

$$\&* := (p, q) \to [p_1 q_1 - \overline{q_2}\, p_2, p_1 q_2 + p_2 \overline{q_1}] \quad (2.3.9.1)$$

```
> p:=[a+I*b,c+I*d]; p&+[0,0]; p&+[-a-I*b,-c-I*d];
```

$$p := [a + \mathrm{I}\, b, c + \mathrm{I}\, d]$$

$$[a + \mathrm{I}\, b, c + \mathrm{I}\, d]$$

$$[0, 0] \quad (2.3.9.2)$$

```
> p&*[1,0]; [1,0]&*p;

  q:=[(a-I*b)/(a^2+b^2+c^2+d^2),(-c-I*d)/(a^2+b^2+c^2+d^2)];
  p&*q;evalc(%);simplify(%); q&*p;evalc(%);simplify(%);
```

$$[a + \mathrm{I}\, b, c + \mathrm{I}\, d]$$

$$[a + \mathrm{I}\, b, c + \mathrm{I}\, d]$$

$$q := \left[\frac{a - \mathrm{I}\, b}{a^2 + b^2 + c^2 + d^2}, \frac{-c - \mathrm{I}\, d}{a^2 + b^2 + c^2 + d^2}\right]$$

$$\left[\frac{(a + \mathrm{I}\, b)(a - \mathrm{I}\, b)}{a^2 + b^2 + c^2 + d^2} - \overline{\frac{-c - \mathrm{I}\, d}{a^2 + b^2 + c^2 + d^2}}\,(c + \mathrm{I}\, d), \frac{(a + \mathrm{I}\, b)(-c - \mathrm{I}\, d)}{a^2 + b^2 + c^2 + d^2} + (c + \mathrm{I}\, d)\,\overline{\frac{a - \mathrm{I}\, b}{a^2 + b^2 + c^2 + d^2}}\right]$$

$$\left[\frac{a^2 + b^2}{a^2 + b^2 + c^2 + d^2} + \frac{c^2}{a^2 + b^2 + c^2 + d^2} + \frac{d^2}{a^2 + b^2 + c^2 + d^2}, \frac{-a\, c + b\, d}{a^2 + b^2 + c^2 + d^2} + \frac{c\, a}{a^2 + b^2 + c^2 + d^2} - \frac{d\, b}{a^2 + b^2 + c^2 + d^2} + \mathrm{I}\left(\frac{-b\, c - a\, d}{a^2 + b^2 + c^2 + d^2} + \frac{d\, a}{a^2 + b^2 + c^2 + d^2} + \frac{c\, b}{a^2 + b^2 + c^2 + d^2}\right)\right]$$

$$[1, 0]$$

$$\left[\frac{(a + \mathrm{I}\, b)(a - \mathrm{I}\, b)}{a^2 + b^2 + c^2 + d^2} - \frac{\overline{c + \mathrm{I}\, d}\,(-c - \mathrm{I}\, d)}{a^2 + b^2 + c^2 + d^2}, \frac{(a - \mathrm{I}\, b)(c + \mathrm{I}\, d)}{a^2 + b^2 + c^2 + d^2} + \frac{(-c - \mathrm{I}\, d)\,\overline{a + \mathrm{I}\, b}}{a^2 + b^2 + c^2 + d^2}\right]$$

$$\left[\frac{a^2+b^2}{a^2+b^2+c^2+d^2}+\frac{c^2+d^2}{a^2+b^2+c^2+d^2}, 0\right]$$
$$[1, 0] \quad (2.3.9.3)$$

```
> z:='z';w:='w';z1:='z1';p:=[z,w];p1:=[z1,w1];p2:=[z2,w2];
  p&*(p1&*p2);expand(%);(p&*p1)&*p2;expand(%);
```

$$z := z$$
$$w := w$$
$$z1 := z1$$
$$p := [z, w]$$
$$p1 := [z1, w1]$$
$$p2 := [z2, w2]$$
$$\left[z\left(z1\,z2-\overline{w2}\,w1\right)-\overline{z1\,w2+w1\,\overline{z2}}\,w,\; z\left(z1\,w2+w1\,\overline{z2}\right)+w\,\overline{z1\,z2-\overline{w2}\,w1}\right]$$
$$\left[z\,z1\,z2-z\,\overline{w2}\,w1-w\,\overline{z1\,w2}-w\,z2\,\overline{w1},\; z\,z1\,w2+z\,w1\,\overline{z2}+w\,\overline{z1\,z2}-w\,w2\,\overline{w1}\right]$$
$$\left[\left(z\,z1-\overline{w1}\,w\right)z2-\overline{w2}\left(z\,w1+w\,\overline{z1}\right),\; \left(z\,z1-\overline{w1}\,w\right)w2+\left(z\,w1+w\,\overline{z1}\right)\overline{z2}\right]$$
$$\left[z\,z1\,z2-w\,z2\,\overline{w1}-z\,\overline{w2}\,w1-\overline{w2}\,w\,\overline{z1},\; z\,z1\,w2-w\,w2\,\overline{w1}+z\,w1\,\overline{z2}+\overline{z2}\,w\,\overline{z1}\right] \quad (2.3.9.4)$$

```
> p&*(p1&+p2);expand(%);(p&*p1)&+(p&*p2);
  (p1&+p2)&*p;expand(%);(p1&*p)&+(p2&*p);
```

$$\left[z\left(z1+z2\right)-\overline{w1+w2}\,w,\; z\left(w1+w2\right)+w\,\overline{z1+z2}\right]$$
$$\left[z\,z1+z\,z2-\overline{w1}\,w-w\,\overline{w2},\; z\,w1+z\,w2+w\,\overline{z1}+w\,\overline{z2}\right]$$
$$\left[z\,z1+z\,z2-\overline{w1}\,w-w\,\overline{w2},\; z\,w1+z\,w2+w\,\overline{z1}+w\,\overline{z2}\right]$$
$$\left[z\left(z1+z2\right)-\overline{w}\left(w1+w2\right),\; \left(z1+z2\right)w+\left(w1+w2\right)\overline{z}\right]$$
$$\left[z\,z1+z\,z2-\overline{w}\,w1-\overline{w}\,w2,\; w\,z1+w\,z2+\overline{z}\,w1+\overline{z}\,w2\right]$$
$$\left[z\,z1+z\,z2-\overline{w}\,w1-\overline{w}\,w2,\; w\,z1+w\,z2+\overline{z}\,w1+\overline{z}\,w2\right] \quad (2.3.9.5)$$

```
> j:=[0,1]; j&*j; [z,0]&+([w,0]&*j);
```

$$j := [0, 1]$$
$$[-1, 0]$$
$$[z, w] \quad (2.3.9.6)$$

```
> k:=[0,I]; k&*k; i:=[I,0]; i&*i; [a,0]&+([b,0]&*i)&+([c,0]&*
  j)&+([d,0]&*k);
```

$$k := [0, I]$$

$$[-1, 0]$$
$$i := [I, 0]$$
$$[-1, 0]$$
$$[a + I\,b,\ c + I\,d] \quad (2.3.9.7)$$

```
> p:=[a+I*b,c+I*d]; evalc([x,0]&*p); evalc(p&*[x,0]);
```

$$p := [a + I\,b,\ c + I\,d]$$
$$[x\,a + I\,x\,b,\ x\,c + I\,x\,d]$$
$$[x\,a + I\,x\,b,\ x\,c + I\,x\,d] \quad (2.3.9.8)$$

```
> j&*[z,0]; [z,0]&*j;
```

$$[0, \bar{z}]$$
$$[0, z] \quad (2.3.9.9)$$

```
> i&*j; j&*k; k&*i; j&*i; k&*j; i&*k;
```

$$[0, I]$$
$$[I, 0]$$
$$[0, 1]$$
$$[0, -I]$$
$$[-I, 0]$$
$$[0, -1] \quad (2.3.9.10)$$

```
> i:='i'; j:='j'; k:='k';

  C2toR4:=q->evalc(Re(q[1])+Im(q[1])*i+Re(q[2])*j+Im(q[2])*k)
  ; q:=C2toR4(p);
```

$$i := i$$
$$j := j$$
$$k := k$$
$$C2toR4 := q \rightarrow evalc\left(\Re(q_1) + \Im(q_1)\,i + \Re(q_2)\,j + \Im(q_2)\,k\right)$$
$$q := a + b\,i + c\,j + d\,k \quad (2.3.9.11)$$

```
> R4toC2:=q->[q-coeff(q,i)*i-coeff(q,j)*j-coeff(q,k)*k+I*
  coeff(q,i),coeff(q,j)+I*coeff(q,k)]; R4toC2(q);
```

$$R4toC2 := q \rightarrow [q - coeff(q, i)\,i - coeff(q, j)\,j - coeff(q, k)\,k + I\,coeff(q, i),\ coeff(q, j) + I\,coeff(q, k)]$$
$$[a + I\,b,\ c + I\,d] \quad (2.3.9.12)$$

```
> qIm:=q->coeff(q,i)*i+coeff(q,j)*j+coeff(q,k)*k;qRe:=q->q-
  qIm(q); qRe(q); qIm(q);
```

$$qIm := q \rightarrow coeff(q, i)\,i + coeff(q, j)\,j + coeff(q, k)\,k$$

$$qRe := q \to q - qIm(q)$$

$$a$$

$$b\,i + c\,j + d\,k \qquad (2.3.9.13)$$

```
> qconjugate:=q->qRe(q)-qIm(q); qconjugate(q);
```

$$qconjugate := q \to qRe(q) - qIm(q)$$

$$a - b\,i - c\,j - d\,k \qquad (2.3.9.14)$$

```
> q; qconjugate(q); qconjugate(%); q+qconjugate(q); q-
  qconjugate(q);
```

$$a + b\,i + c\,j + d\,k$$

$$a - b\,i - c\,j - d\,k$$

$$a + b\,i + c\,j + d\,k$$

$$2\,a$$

$$2\,b\,i + 2\,c\,j + 2\,d\,k \qquad (2.3.9.15)$$

```
> q1:=a1+b1*i+c1*j+d1*k; q2:=a2+b2*i+c2*j+d2*k;

  q1+q2; collect(%,[i,j,k]); `&+`:=(q1,q2)->collect(q1+q2,[i,
  j,k]); q1&+q2;

  `&*`:=proc(q1,q2) local a1,a2,b1,b2,c1,c2,d1,d2;
  a1:=qRe(q1);a2:=qRe(q2);b1:=coeff(q1,i);b2:=coeff(q2,i);
  c1:=coeff(q1,j);c2:=coeff(q2,j);d1:=coeff(q1,k);d2:=coeff
  (q2,k);
  (a1*a2-b1*b2-c1*c2-d1*d2)+(a1*b2+a2*b1+c1*d2-d1*c2)*i+
  (a1*c2+c1*a2+d1*b2-b1*d2)*j+(a1*d2+d1*a2+b1*c2-c1*b2)*k;
  end;

  q1&*q2;

  qconjugate(q1&+q2); qconjugate(q1)&+qconjugate(q2);

  qconjugate(q1&*q2);qconjugate(q2)&*qconjugate(q1); expand
  (%%-%);
```

$$q1 := a1 + b1\,i + c1\,j + d1\,k$$

$$q2 := a2 + b2\,i + c2\,j + d2\,k$$

$$a1 + b1\,i + c1\,j + d1\,k + a2 + b2\,i + c2\,j + d2\,k$$

$$(b1 + b2)\,i + (c2 + c1)\,j + (d2 + d1)\,k + a1 + a2$$

$$\&+ := (q1, q2) \to collect(q1 + q2, [i, j, k])$$

$$(b1 + b2)\,i + (c2 + c1)\,j + (d2 + d1)\,k + a1 + a2$$

$\&* :=$ **proc**$(q1, q2)$

local $a1, a2, b1, b2, c1, c2, d1, d2$;

$a1 := qRe(q1)$;

$a2 := qRe(q2);$
$b1 := coeff(q1, i);$
$b2 := coeff(q2, i);$
$c1 := coeff(q1, j);$
$c2 := coeff(q2, j);$
$d1 := coeff(q1, k);$
$d2 := coeff(q2, k);$
$a1*a2 - b1*b2 - c1*c2 - d1*d2 + (a1*b2 + a2*b1 + c1*d2 - d1*c2)*i + (a1*c2 + c1*a2 + d1*b2 - b1*d2)*j + (a1*d2 + d1*a2 + b1*c2 - c1*b2)*k$

end proc

$$a1\,a2 - b1\,b2 - c1\,c2 - d1\,d2 + (a1\,b2 + a2\,b1 + c1\,d2 - d1\,c2)\,i + (a1\,c2 + c1\,a2 + d1\,b2 - b1\,d2)\,j + (a1\,d2 + d1\,a2 + b1\,c2 - c1\,b2)\,k$$

$$a1 + a2 - (b1 + b2)\,i - (c2 + c1)\,j - (d2 + d1)\,k$$

$$(-b1 - b2)\,i + (-c2 - c1)\,j + (-d2 - d1)\,k + a1 + a2$$

$$a1\,a2 - b1\,b2 - c1\,c2 - d1\,d2 - (a1\,b2 + a2\,b1 + c1\,d2 - d1\,c2)\,i - (a1\,c2 + c1\,a2 + d1\,b2 - b1\,d2)\,j - (a1\,d2 + d1\,a2 + b1\,c2 - c1\,b2)\,k$$

$$a1\,a2 - b1\,b2 - c1\,c2 - d1\,d2 + (-a2\,b1 - a1\,b2 + d1\,c2 - c1\,d2)\,i + (-c1\,a2 - a1\,c2 + b1\,d2 - d1\,b2)\,j + (-d1\,a2 - a1\,d2 + c1\,b2 - b1\,c2)\,k$$

$$0 \qquad (2.3.9.16)$$

2.3.10. Kvaterniók és a három dimenziós euklidészi tér.

```
> q1:=b1*i+c1*j+d1*k; q2:=b2*i+c2*j+d2*k; q3:=b3*i+c3*j+d3*k;

  scalarprod:=(q1,q2)->-qRe(q1&*q2); scalarprod(q1,q2);

  vectorprod:=(q1,q2)->qIm(q1&*q2); vectorprod(q1,q2);

  mixedprod:=(q1,q2,q3)->scalarprod(q1,vectorprod(q2,q3));
  mixedprod(q1,q2,q3);
```

$$q1 := b1\,i + c1\,j + d1\,k$$

$$q2 := b2\,i + c2\,j + d2\,k$$

$$q3 := b3\,i + c3\,j + d3\,k$$

$$scalarprod := (q1, q2) \rightarrow -qRe(q1 \,\&*\, q2)$$

$$b1\,b2 + c1\,c2 + d1\,d2$$

$$vectorprod := (q1, q2) \rightarrow qIm(q1 \&^* q2)$$
$$(c1\, d2 - d1\, c2)\, i + (d1\, b2 - b1\, d2)\, j + (b1\, c2 - c1\, b2)\, k$$
$$mixedprod := (q1, q2, q3) \rightarrow scalarprod(q1, vectorprod(q2, q3))$$
$$b1\,(c2\, d3 - d2\, c3) + c1\,(d2\, b3 - b2\, d3) + d1\,(b2\, c3 - c2\, b3) \qquad (2.3.10.1)$$

2.3.11. Kvaterniók abszolút értéke.

```
> qabs:=q->sqrt(qRe(q)^2+coeff(q,i)^2+coeff(q,j)^2+coeff(q,k)
^2);

qabs(q); q&*qconjugate(q);
```

$$qabs := q \rightarrow \sqrt{qRe(q)^2 + coeff(q, i)^2 + coeff(q, j)^2 + coeff(q, k)^2}$$
$$\sqrt{a^2 + b^2 + c^2 + d^2}$$
$$a^2 + b^2 + c^2 + d^2 \qquad (2.3.11.1)$$

2.3.12. A skalár- vektor- és vegyes szorzat geometriai jelentése.

```
> qabs(q1&+q2)^2-qabs(q1-q2)^2;
```

$$4\, b1\, b2 + 4\, c1\, c2 + 4\, d1\, d2 \qquad (2.3.12.1)$$

2.4. Polinomok

```
> restart;
```

2.4.1. Jelölés.

2.4.2. Polinomok és racionális törtfüggvények.

2.4.3. A maradékos osztás tétele polinomokra.

```
> f:=3*x^3+6*x^2+7*x-8; g:=x^2+8*x-2; q:=quo(f,g,x); r:=rem
(f,g,x);
g*q+r; expand(%);
```

$$f := 3\, x^3 + 6\, x^2 + 7\, x - 8$$
$$g := x^2 + 8\, x - 2$$
$$q := 3\, x - 18$$
$$r := -44 + 157\, x$$
$$(x^2 + 8\, x - 2)\,(3\, x - 18) - 44 + 157\, x$$
$$3\, x^3 + 6\, x^2 + 7\, x - 8 \qquad (2.4.3.1)$$

2.4.4. Következmény: Horner-elrendezés.

```
> rem(f,x-5,x); subs(x=5,f);
```

$$552$$

$$552 \tag{2.4.4.1}$$

2.4.5. Következmény: gyöktényező leválasztása.

```
> f:=(x-1)^2*(x-2); f:=expand(f); quo(f,x-2,x);
```

$$f := (x-1)^2 (x-2)$$

$$f := x^3 - 4x^2 + 5x - 2$$

$$x^2 - 2x + 1 \tag{2.4.5.1}$$

2.4.6. Következmény.

2.4.7. Következmény.

2.4.8. Polinomok egyértelmű felírása.

```
> coeff(f,x,0); coeff(f,x,1); coeff(f,x,2); coeff(f,x,3);
  coeff(f,x,4); lcoeff(f,x); lcoeff(0,x);
```

$$-2$$

$$5$$

$$-4$$

$$1$$

$$0$$

$$1$$

$$0 \tag{2.4.8.1}$$

```
> degree(f); degree(0);
```

$$3$$

$$-\infty \tag{2.4.8.2}$$

2.4.9. A maradékos osztás egyértelműsége.

2.4.10. Többszörös gyökök.

```
> subs(x=1,f); quo(f,x-1,x); subs(x=1,f); quo(f,(x-1)^2,x);
```

$$0$$

$$x^2 - 3x + 2$$

$$0$$
$$x-2 \qquad (2.4.10.1)$$

2.4.11. *Az algebra alaptétele.*

```
> solve(f,x); solve(x^3=1,x); r:=[%];
```

$$2, 1, 1$$
$$1, -\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\sqrt{3}, -\frac{1}{2}-\frac{1}{2}\,\mathrm{I}\sqrt{3}$$
$$r := \left[1, -\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\sqrt{3}, -\frac{1}{2}-\frac{1}{2}\,\mathrm{I}\sqrt{3}\right] \qquad (2.4.11.1)$$

2.4.12. *Gyöktényezős előállítás.*

```
> map(y->x-y,r); convert(%,`*`); evalc(%);
```

$$\left[x-1, x+\frac{1}{2}-\frac{1}{2}\,\mathrm{I}\sqrt{3}, x+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\sqrt{3}\right]$$
$$(x-1)\left(x+\frac{1}{2}-\frac{1}{2}\,\mathrm{I}\sqrt{3}\right)\left(x+\frac{1}{2}+\frac{1}{2}\,\mathrm{I}\sqrt{3}\right)$$
$$x^3-1 \qquad (2.4.12.1)$$

* 2.4.13. *Lagrange-interpoláció.*

```
> CurveFitting[PolynomialInterpolation]([0,1,2,3],[0,3,1,3],
  x);
  CurveFitting[PolynomialInterpolation]([0,1,2,3],[0,3,1,3],
  x,form=Lagrange);
```

$$\frac{3}{2}\,x^3-7\,x^2+\frac{17}{2}\,x$$
$$\frac{3}{2}\,x\,(x-2)\,(x-3)-\frac{1}{2}\,x\,(x-1)\,(x-3)+\frac{1}{2}\,x\,(x-1)\,(x-2) \qquad (2.4.13.1)$$

2.4.14. *Többváltozós polinomok és racionális törtfüggvények.*

```
> f:=5*x^2*y; type(f,monomial); type(f,polynom); degree(f);
  f:=f+x*y^3; type(f,monomial); type(f,polynom); degree(f);
```

$$f := 5\,x^2\,y$$

$$true$$

$$true$$

$$3$$

$$f := 5\,x^2\,y + x\,y^3$$

$$false$$

$$true$$

$$4 \qquad (2.4.14.1)$$

```
> f/(2*x*y-5*x^2*y^2); type(f,ratpoly);
```

$$\frac{5\,x^2\,y + x\,y^3}{2\,x\,y - 5\,x^2\,y^2}$$

$$true \qquad (2.4.14.2)$$

▶ 3. Határérték

▶ 4. Differenciálszámítás

▶ 5. Integrálszámítás